# Blatt 8

## Aufgabe 30

```
> restart:
> with(LinearAlgebra):
> A := <<1|1|4>,<1|-3|2>,<1|2|-1>>;
```

$$A := \begin{bmatrix} 1 & 1 & 4 \\ 1 & -3 & 2 \\ 1 & 2 & -1 \end{bmatrix} \quad (1.1)$$

```
> B := <<7|8|9>, <4|5|6>, <1|2|3>>;
```

$$B := \begin{bmatrix} 7 & 8 & 9 \\ 4 & 5 & 6 \\ 1 & 2 & 3 \end{bmatrix} \quad (1.2)$$

```
> b := <-1, 0, 1>;
```

$$b := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \quad (1.3)$$

```
> Rank(A);
```

$$3 \quad (1.4)$$

```
> # Die Matrix A hat Rang 3. Daher ist das Gleichungssystem Ax=b
  eindeutig loesbar, denn es gibt genau eine Loesung.
```

```
> Rank(B);
```

$$2 \quad (1.5)$$

```
> # Die Matrix B hat Rang 2. Daher besitzt das Gleichungssystem
  Bx=b mehrere oder keine Loesungen.
```

```
> # Zeilenstufenformen der Matrizen <A|b>,<B|b>
> G1 := <A|b>;
```

$$G1 := \begin{bmatrix} 1 & 1 & 4 & -1 \\ 1 & -3 & 2 & 0 \\ 1 & 2 & -1 & 1 \end{bmatrix} \quad (1.6)$$

```
> G2 := <B|b>;
```

(1.7)

$$G2 := \begin{bmatrix} 7 & 8 & 9 & -1 \\ 4 & 5 & 6 & 0 \\ 1 & 2 & 3 & 1 \end{bmatrix} \qquad (1.7)$$

```
> ZA := ReducedRowEchelonForm(G1);
```

$$ZA := \begin{bmatrix} 1 & 0 & 0 & \frac{15}{22} \\ 0 & 1 & 0 & -\frac{1}{22} \\ 0 & 0 & 1 & -\frac{9}{22} \end{bmatrix} \qquad (1.8)$$

```
> ZB := ReducedRowEchelonForm(G2);
```

$$ZB := \begin{bmatrix} 1 & 0 & -1 & -\frac{5}{3} \\ 0 & 1 & 2 & \frac{4}{3} \\ 0 & 0 & 0 & 0 \end{bmatrix} \qquad (1.9)$$

```
> # Loesen der Gleichungssysteme
> x_a := LinearSolve(A, b);
```

$$x\_a := \begin{bmatrix} \frac{15}{22} \\ -\frac{1}{22} \\ -\frac{9}{22} \end{bmatrix} \qquad (1.10)$$

```
> x_b := LinearSolve(B, b);
```

$$x\_b := \begin{bmatrix} -\frac{5}{3} + \_t0_3 \\ \frac{4}{3} - 2\,\_t0_3 \\ \_t0_3 \end{bmatrix} \qquad (1.11)$$

```
> # Ueberpruefen der Ergebnisse durch Einsetzen
> A . x_a;
```

(1.12)

$$\begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \quad (1.12)$$

```
> B . x_b;
```

$$\begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix} \quad (1.13)$$

## Aufgabe 31

```
> restart:
> with(plots):
> with(LinearAlgebra):
```

```
> dreh := <<3, 1> | <1, 4>>;
```

$$dreh := \begin{bmatrix} 3 & 1 \\ 1 & 4 \end{bmatrix} \quad (2.1)$$

```
> # Sei y in Bild(f). Dann gilt y = A*x fuer ein x aus K. Da A
  invertierbar, gilt x = A^(-1)*y.
> # Somit folgt y = A*( A^(-1)*y ), wobei A^(-1)*y in K, also
  Norm 1 hat.
```

```
> G := dreh^(-1) . <x,y>;
```

$$G := \begin{bmatrix} \frac{4x}{11} - \frac{y}{11} \\ -\frac{x}{11} + \frac{3y}{11} \end{bmatrix} \quad (2.2)$$

```
> implicitplot(VectorNorm(G,2)=1,x=-5..5,y=-5..5,numpoints=
  100000);
```




```
> # mit Parametrisierung
> K := plot([ cos(t), sin(t), t=0..2*Pi ],color=red,numpoints=
  2000):
> par := < cos(t), sin(t) >;
```

$$par := \begin{bmatrix} \cos(t) \\ \sin(t) \end{bmatrix} \tag{2.3}$$

```
> E1 := dreh.par;
```

$$E1 := \begin{bmatrix} 3\cos(t) + \sin(t) \\ \cos(t) + 4\sin(t) \end{bmatrix} \tag{2.4}$$

```
> E_1 := plot([E1[1],E1[2],t=0..2*Pi],color=green,numpoints=3000)
  :
> display({ K, E_1 }, numpoints = 2000, scaling = constrained);
```




# Aufgabe 32

```
> restart:
> with(plots):
> with(LinearAlgebra):
```

(a)

```
> A := << 4, 2, 1 > | < 2, 3, 1 > | < 1, 1, 2 >>;
  #A := << 4, 1, 0 > | < 1, 3, 0 > | < 0, 0, 1 >>;
```

$$A := \begin{bmatrix} 4 & 2 & 1 \\ 2 & 3 & 1 \\ 1 & 1 & 2 \end{bmatrix} \qquad (3.1)$$

```
> # Gleichung
> G := expand(Transpose(A^(-1) . < x, y, z >).(A^(-1) . < x, y, z >));
```

$$G := \frac{35}{169}\,x^2 - \frac{68}{169}\,x\,y - \frac{14}{169}\,x\,z + \frac{62}{169}\,y^2 - \frac{54}{169}\,y\,z + \frac{69}{169}\,z^2 \qquad (3.2)$$

```
> # Parametrisierung
> B := A . < cos(t) * sin(s), sin(t) * sin(s), cos(s) >;
```

$$B := \begin{bmatrix} 4\cos(t)\sin(s) + 2\sin(t)\sin(s) + \cos(s) \\ 2\cos(t)\sin(s) + 3\sin(t)\sin(s) + \cos(s) \\ \cos(t)\sin(s) + \sin(t)\sin(s) + 2\cos(s) \end{bmatrix} \quad (3.3)$$

```
> plot3d([ B[1], B[2], B[3] ], t = 0..2*Pi, s = 0..Pi, numpoints
  = 3000, style = wireframe, shading = zgrayscale, axes = boxed,
  scaling=constrained);
```




```
> implicitplot3d(G = 1, x = -5..5, y = -5..5, z = -3..3,numpoints
  = 5000, style = wireframe, shading = zgrayscale, axes = boxed,
  scaling = constrained);
```




(b)

```
> E_W, E_V := Eigenvectors(A):
> # Test
> for kk from 1 to 3 do
    simplify(E_W[kk] * E_V[1..-1, kk] - A . E_V[1..-1, kk]);
  end do;
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

**(3.4)**

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{3.4}$$

```
> # Etwas kompakter vielleicht?
> EW := map(simplify@evalc, E_W);
> EV := map(simplify@evalc, E_V):
```

$$EW := \begin{bmatrix} 3 + \frac{2\sqrt{3}\sqrt{7}\cos\left(\frac{\arctan\left(\frac{\sqrt{3}}{9}\right)}{3}\right)}{3} \\ 3 - \sqrt{7}\sin\left(\frac{\arctan\left(\frac{\sqrt{3}}{9}\right)}{3}\right) - \frac{\sqrt{3}\sqrt{7}\cos\left(\frac{\arctan\left(\frac{\sqrt{3}}{9}\right)}{3}\right)}{3} \\ 3 + \sqrt{7}\sin\left(\frac{\arctan\left(\frac{\sqrt{3}}{9}\right)}{3}\right) - \frac{\sqrt{3}\sqrt{7}\cos\left(\frac{\arctan\left(\frac{\sqrt{3}}{9}\right)}{3}\right)}{3} \end{bmatrix} \tag{3.5}$$

```
> # Die aufgespannten Eigenräume
> V1 := t * (EV[1..-1, 1] / Norm(EV[1..-1, 1], 2)):
> V2 := t * (EV[1..-1, 2] / Norm(EV[1..-1, 2], 2)):
> V3 := t * (EV[1..-1, 3] / Norm(EV[1..-1, 3], 2)):
```

```
> P1 := plot3d([ B[1], B[2], B[3] ], t = 0..2*Pi, s = 0..Pi,
  numpoints = 3000, style = wireframe, axes = boxed, scaling=
  constrained);
```

$$P1 := PLOT3D(...) \tag{3.6}$$

```
> # Eigenräume einzeichnen
> P2 := spacecurve(Matrix(V1), t = 0 .. EW[1], thickness = 2,
  colour = red);
> P3 := spacecurve(Matrix(V2), t = 0 .. EW[2], thickness = 2,
  colour = blue);
> P4 := spacecurve(Matrix(V3), t = 0 .. EW[3], thickness = 2,
  colour = green);
```

$$P2 := PLOT3D(...)$$
$$P3 := PLOT3D(...)$$
$$P4 := PLOT3D(...) \tag{3.7}$$

```
> Matrix(V1):
> display(P1, P2, P3, P4);
```




# Aufgabe 33

```
> restart:
```

```
> f := cos(x) / x;
```

$$f := \frac{\cos(x)}{x} \quad (4.1)$$

```
> G := Int(f, x);
```

$$G := \int \frac{\cos(x)}{x}\, dx \quad (4.2)$$

```
> G = value(G);
```

$$\int \frac{\cos(x)}{x}\, dx = \mathrm{Ci}(x) \quad (4.3)$$

```
> F := simplify(value(G));
```

$$F := \mathrm{Ci}(x) \quad (4.4)$$

```
> Tf:= series(f, x = 0, 12);
```

$$Tf := x^{-1} - \frac{1}{2}\,x + \frac{1}{24}\,x^3 - \frac{1}{720}\,x^5 + \frac{1}{40320}\,x^7 - \frac{1}{3628800}\,x^9 + \mathrm{O}(x^{11}) \tag{4.5}$$

```
> Pf := convert(Tf, polynom);
```

$$Pf := \frac{1}{x} - \frac{x}{2} + \frac{x^3}{24} - \frac{x^5}{720} + \frac{x^7}{40320} - \frac{x^9}{3628800} \tag{4.6}$$

```
> TF := series(F, x = 0, 12);
```

$$TF := \gamma + \ln(x) - \frac{1}{4}\,x^2 + \frac{1}{96}\,x^4 - \frac{1}{4320}\,x^6 + \frac{1}{322560}\,x^8 - \frac{1}{36288000}\,x^{10} + \mathrm{O}(x^{12}) \tag{4.7}$$

```
> PF := convert(TF, polynom);
```

$$PF := \gamma + \ln(x) - \frac{x^2}{4} + \frac{x^4}{96} - \frac{x^6}{4320} + \frac{x^8}{322560} - \frac{x^{10}}{36288000} \tag{4.8}$$

```
> Abl_Taylor := diff(PF, x);
```

$$Abl\_Taylor := \frac{1}{x} - \frac{x}{2} + \frac{x^3}{24} - \frac{x^5}{720} + \frac{x^7}{40320} - \frac{x^9}{3628800} \tag{4.9}$$

```
> vergleich := Abl_Taylor - Pf;
```

$$vergleich := 0 \tag{4.10}$$